Integrazione numerica mediante polinomi interpolatori

Note

Autore

Claudio Marsan
Liceo Cantonale di Mendrisio
Via Agostino Maspoli
CH-6850 Mendrisio (Switzerland)
e-mail: claudio.marsan@liceomendrisio.ch

Versione

Versione 2.0, 16 marzo 2003
Maple V Release 6.02 for Windows 2000

```
> restart:
> with(plots):
Warning, the name changecoords has been redefined
> with(plottools):
```

Definizione degli estremi dell'intervallo di integrazione

```
> a := Pi/6; b := 3*Pi/2;
```

$$a := \frac{1}{6}\pi$$

$$b := \frac{3}{2}\pi$$

La funzione da integrare

```
> f := x -> cos(x) + 0.5 * cos(3*x) + 3;
```

$$f := x \to \cos(x) + .5\cos(3x) + 3$$

Un po' di "cosmesi" per la grafica

```
> line_a := line([a,0], [a,f(a)], color=black, linestyle=3,
  thickness=3):
> line_b := line([b,0], [b,f(b)], color=black, linestyle=3,
  thickness=3):
> gr_a := plots[display](line_a):
> gr_b := plots[display](line_b):
> gr_f := plot(f(x), x=0..6, color=red, thickness=3):
> display({gr_a, gr_b, gr_f});
```



Esempio 1.

Approssimazione con il polinomio costante p(x) = f(x1), dove x1 è il punto medio dell'intervallo [a, b]

```
> x1 := 0.5*(a + b);
```

$$x1 := .8333333333\,\pi$$

```
> p := x -> evalf(f(x1));
```

$$p := x \rightarrow \text{evalf}(\,\text{f}(x1\,)\,)$$

```
> gr_p := plot(p(x), x=0..6, color=blue, thickness=3):
> line_x1 := line([x1,0], [x1,p(x1)], color=black, linestyle=3):
> gr_x1 := plots[display](line_x1):
> display({gr_a, gr_b, gr_f, gr_p, gr_x1});
```




```
> A := integrate(f(x), x=a..b); # valore esatto
```

$$A := 11.06637062$$

```
> A1 := integrate(p(x), x=a..b);
```

$$A1 := 8.938771885$$

```
> errore1 := evalf((A - A1)/A)*100; # errore percentuale
```

$$errore1 := 19.22580409$$

Esempio 2.

Approssimazione con il polinomio di Lagrange q(x) di grado 3 passante per i punti (x1, f(x1)), (x2, f(x2)), (x3, f(x3)) e (x4, f(x4)) , dove x1= $\frac{\pi}{6}$, x2 = x1 + 1, x3 = x1 + 2 e x4 = x1 + 3.

```
> x1 := a; x2 := x1 + 1; x3 := x1 + 2; x4 := x1 + 3;
```

$$x1 := \frac{1}{6}\pi$$

$$x2 := \frac{1}{6}\pi + 1$$

$$x3 := \frac{1}{6}\pi + 2$$

$$x4 := \frac{1}{6}\pi + 3$$

```
> X := evalf([x1, x2, x3, x4]); # ascisse
```

$$X := [.5235987758, 1.523598776, 2.523598776, 3.523598776]$$

```
> Y := map(f, X); # ordinate
```

$$Y := [3.866025404, 2.976620026, 2.324665303, 1.866022103]$$

Funzione per calcolare il polinomio di interpolazione di Lagrange:

```
> Lagrange := proc(x, ax, ay)
    local value, i, n, j, prod;
    value := 0;
    n := nops(ax) - 1;
    for i from 0 to n do
      prod := 1;
      for j from 0 to n do
        if j<>i then
           prod := prod*(x-ax[j+1])/(ax[i+1]-ax[j+1]);
        end if;
      end do;
      value := value + ay[i+1] *prod;
    end do;
  end;
```

Lagrange := **proc**(x, ax, ay)

local *value*, *i*, *n*, *j*, *prod*;

$value := 0$;

$n := \mathrm{nops}(ax) - 1$;

for i **from** 0 **to** n **do**

prod := 1;

for j **from** 0 **to** n **do**

if $j \neq i$ **then** $prod := prod*(x - ax[j+1]) / (ax[i+1] - ax[j+1])$ **end if**

end do;

$value := value + ay[i+1]*prod$

end do

end proc

```
> q := x -> sort(expand(Lagrange(x, X, Y)));
```

$$q := x \rightarrow \mathrm{sort}(\mathrm{expand}(\mathrm{Lagrange}(x, X, Y)))$$

```
> q(x);
```

$$-.007356523\, x^3 + .152350493\, x^2 - 1.17633432\, x + 4.441240870$$

```
> gr_q := plot(q(x), x=0..6, color=blue, thickness=3):
> line_x1 := line([x1,0], [x1,f(x1)], color=black, linestyle=3):
> gr_x1 := plots[display](line_x1):
> line_x2 := line([x2,0], [x2,f(x2)], color=black, linestyle=3):
> gr_x2 := plots[display](line_x2):
> line_x3 := line([x3,0], [x3,f(x3)], color=black, linestyle=3):
```

```
> gr_x3 := plots[display](line_x3):
> line_x4 := line([x4,0], [x4,f(x4)], color=black, linestyle=3):
> gr_x4 := plots[display](line_x4):
> display({gr_a, gr_b, gr_f, gr_q, gr_x1, gr_x2, gr_x3, gr_x4});
```




```
> A := integrate(f(x), x=a..b); # valore esatto
```

$$A := 11.06637062$$

```
> A2 := integrate(q(x), x=a..b);
```

$$A2 := 10.10368753$$

```
> errore2 := evalf((A - A2)/A)*100; # errore percentuale
```

$$errore2 := 8.699176298$$

Esempio 3.

Approssimazione con il polinomio di Lagrange r(x) di grado 4 passante per i punti (x1, f(x1)), (x2, f(x2)), (x3, f(x3)), (x4, f(x4)) e (x5, f(x5)), dove x1=$\frac{\pi}{6}$, x2 = x1 + 1, x3 = x1 + 2, x4 = x1 + 3 e x5 = $\frac{3\,\pi}{2}$.

```
> x1 := a; x2 := x1 + 1; x3 := x1 + 2; x4 := x1 + 3; x5 := b;
```

$$x1 := \frac{1}{6}\pi$$

$$x2 := \frac{1}{6}\pi + 1$$

$$x3 := \frac{1}{6}\pi + 2$$

$$x4 := \frac{1}{6}\pi + 3$$

$$x5 := \frac{3}{2}\pi$$

```
> X := evalf([x1, x2, x3, x4, x5]); # ascisse
```

$$X := [.5235987758, 1.523598776, 2.523598776, 3.523598776, 4.712388981]$$

```
> Y := map(f, X); # ordinate
```

$$Y := [3.866025404, 2.976620026, 2.324665303, 1.866022103, 3.000000001]$$

```
> r := x -> sort(expand(Lagrange(x, X, Y)));
```

$$r := x \rightarrow \mathrm{sort}(\mathrm{expand}(\mathrm{Lagrange}(x, X, Y)))$$

```
> r(x);
```

$$.04283485358\, x^4 - .354078751\, x^3 + 1.097703385\, x^2 - 2.162742447\, x + 4.745100696$$

```
> gr_r := plot(r(x), x=0..6, y=0..6, color=blue, thickness=3):
> line_x5 := line([x5,0], [x5,f(x5)], color=black, linestyle=3):
> gr_x5 := plots[display](line_x5):
> display({gr_a, gr_b, gr_f, gr_r, gr_x1, gr_x2, gr_x3, gr_x4,
gr_x5});
```




```
> A := integrate(f(x), x=a..b); # valore esatto
```

$$A := 11.06637062$$

```
> A3 := integrate(r(x), x=a..b);
```

$$A3 := 10.65927294$$

```
> errore3 := evalf((A - A3)/A)*100; # errore percentuale
```

$$errore3 := 3.678691903$$

Integrazione numerica mediante archi di curva (o a tratti)

Esempio 4.

Applichiamo il metodo dell'esempio 1 agli intervalli definiti nell'esempio 3.

```
> x12 := 0.5*(x1 + x2);
```

$$x12 := .1666666667\,\pi + .5$$

```
> line_x12 := line([x12,0], [x12,f(x12)], color=blue, linestyle=2):
> gr_x12 := plots[display](line_x12):
> s1 := x -> f(x12); gr_s1 := plot(s1(x), x=x1..x2, color=blue, thickness=3):
```

$$s1 := x \to f(x12)$$

```
> x23 := 0.5*(x2 + x3);
```

$$x23 := .1666666667\,\pi + 1.5$$

```
> line_x23 := line([x23,0], [x23,f(x23)], color=blue, linestyle=2):
> gr_x23 := plots[display](line_x23):
> s2 := x -> f(x23); gr_s2 := plot(s2(x), x=x2..x3, color=blue, thickness=3):
```

$$s2 := x \to f(x23)$$

```
> x34 := 0.5*(x3 + x4);
```

$$x34 := .1666666667\,\pi + 2.5$$

```
> line_x34 := line([x34,0], [x34,f(x34)], color=blue, linestyle=2):
> gr_x34 := plots[display](line_x34):
> s3 := x -> f(x34); gr_s3 := plot(s3(x), x=x3..x4, color=blue, thickness=3):
```

$$s3 := x \to f(x34)$$

```
> x45 := 0.5*(x4 + x5);
```

$$x45 := .8333333333\,\pi + 1.5$$

```
> line_x45 := line([x45,0], [x45,f(x45)], color=blue, linestyle=2):
> gr_x45 := plots[display](line_x45):
```

```
> s4 := x -> f(x45); gr_s4 := plot(s4(x), x=x4..x5, color=blue,
  thickness=3):
```

$$s4 := x \to \mathrm{f}(x45)$$

```
> line_x1 := line([x1,0], [x1,s1(x1)], color=black, linestyle=3):
> gr_x1 := plots[display](line_x1):
> line_x2 := line([x2,0], [x2,max(s1(x2),s2(x2))], color=black,
  linestyle=3):
> gr_x2 := plots[display](line_x2):
> line_x3 := line([x3,0], [x3,max(s2(x3),s3(x3))], color=black,
  linestyle=3):
> gr_x3 := plots[display](line_x3):
> line_x4 := line([x4,0], [x4,max(s3(x4),s4(x4))], color=black,
  linestyle=3):
> gr_x4 := plots[display](line_x4):
> display({gr_a, gr_b, gr_f, gr_s1, gr_s2, gr_s3, gr_s4, gr_x1,
  gr_x2, gr_x3, gr_x4, gr_x5, gr_x12, gr_x23, gr_x34, gr_x45});
```




```
> A := integrate(f(x), x=a..b); # valore esatto
```

$$A := 11.06637062$$

```
> A4 := integrate(s1(x), x=x1..x2) + integrate(s2(x), x=x2..x3) +
  integrate(s3(x), x=x3..x4) + integrate(s4(x), x=x4..x5);
```

$$A4 := 11.09245758$$

```
> errore4 := evalf((A - A4)/A)*100; # errore percentuale
```

$$errore4 := -.2357318483$$

Esempio 5.

Come nell'esempio 4, ma considerando 8 intervalli.

```
> X := [a, a + 0.5, a + 1, a + 1.5, a + 2, a + 2.5, a + 3, a +
  3.5, b];
```

$$X := \left[\frac{1}{6}\pi, \frac{1}{6}\pi + .5, \frac{1}{6}\pi + 1, \frac{1}{6}\pi + 1.5, \frac{1}{6}\pi + 2, \frac{1}{6}\pi + 2.5, \frac{1}{6}\pi + 3, \frac{1}{6}\pi + 3.5, \frac{3}{2}\pi\right]$$

```
> A := integrate(f(x), x=a..b); # valore esatto
```

$$A := 11.06637062$$

```
> A5 := evalf(sum((X[k+1] - X[k])*f(0.5*(X[k] + X[k+1])),
  k=1..8));
```

$$A5 := 11.07367977$$

```
> errore4 := evalf((A - A5)/A)*100; # errore percentuale
```

$$errore4 := -.06604830302$$